

10.1 Qbasic'te Grafik

Qbasic'te çıktı ekranı iki şekilde kullanılır: Birincisi, bu bölüme kadar yaptığımız tüm işler için kullandığımız **yazı** (text) modu, diğeri ise **grafik** modudur. Grafik modu, ekrana çeşitli şekillerin çizdirilmesi ve bunların renklendirilmesi gibi işlerin yapıldığı moddur. Bu bölümde yapacağımız tüm işler için, çıktı ekranını grafik modunda kullanacağız.


● Ekran Modunun Belirlenmesi – SCREEN Deyimi

Çıktı ekranının hangi modda çalışacağını belirlemek için SCREEN deyimi kullanılır.

Dizilimi: SCREEN *mod, renk seçeneği*

Mod: On dört (0~13) değişik değer alabilen tam sayı, değişken veya ifadeden oluşur.

Örnek:





Dizilimdeki “2” değeri ekran modunu , “1” değeri ise ekrandaki yazıların renkli olup olmasını belirleyen renk kodudur. Ekran modlarının seçimi ile ekran üzerindeki nokta (pixel) sayısı ve ekranın metin kapasitesi de birlikte seçilmiş olur. SCREEN deyimi ile kullanılacak ekran modları, nokta sayısı ve metin kapasiteleri aşağıda tablo hâlinde verilmiştir:

Mod	Nokta (çözünürlük)	Satır x Sütun
0 (yazı)	-	25x40 veya 25x80
1,7,13	320x200	25x40
2,8	640x200	25x80
9,10	640x350	25x80
11,12	640x480	30x80 60x80

3, 4, 5 ve 6 modları, renkli olmayan ekran kartları için geçerlidir. VGA veya EGA görüntü kartı olan bir bilgisayarda kullanılırsa hata mesajı verir. Renksiz ekranlar artık kullanılmadığı için bu modların kullanımına değinmeyeceğiz.

Modlara göre ekran yoğunluğu değişmektedir. Bu noktalar grafik şekillerinin elde edilmesi için grafik adaptörünün ışıklandığı noktalardır. Örneğin bir çizginin 20 veya 40 noktadan oluşması, netliğini ve görünümünü değiştirir. Çizgi, 40 nokta ile elde edildiğinde daha net ve daha düzgün görünecektir. VGA (Video Graphic Adapter) grafik kartına sahip bir bilgisayarda, ekrandaki yatay ve düşey nokta sayıları aşağıdaki gibi olacaktır:



● COLOR Deyimi

Grafik modunda COLOR deyimi, zemin renginin ve çizilecek grafik şekillerinin renklerinin belirlenmesinde kullanılır. COLOR deyiminin, her SCREEN modu için değişik kullanımı vardır. COLOR deyimini daha önce 8. bölümde incelemiştik. Burada tekrar ele almamızın sebebi, deyimin grafik modu kullanımının farklılıklar içermesidir.

Dizilim:

COLOR Fon (Zemin rengi), Palet

Fon (zemin) rengi, 8. bölümde anlatılan renk ve renk kodlarını içerir. Palet rengi ise 1 veya 0 olabilir.

Örnek:

COLOR 2, 1

Örnek dizilimde verilen zemin rengi yeşildir. Palet 1 olduğundan grafik modunda çizilecek nesnelerin renkleri (çizimi gerçekleştiren deyimlerde kullanılan renk kodlarına göre) turkuvaz, mor veya beyaz olabilir.

Seçilen palet koduna göre elde edilebilecek renkler, aşağıdaki tabloda verilmiştir:

Renk kodu	Palet: 0	Palet: 1
1	Yeşil	Turkuvaz
2	Kırmızı	Mor
3	Sarı	Beyaz

• LINE Deyimi

LINE deyimi ekrana çizgi çizdirmek için kullanılır. Ekranda iki noktanın koordinatları belirlenir ve bu iki noktayı birleştiren çizgi çizilir. SCREEN deyiminde öğrendiğimiz noktalar, çizgilerin başlangıç ve bitiş koordinatlarını belirtir.

Dizilim:

```
LINE (X, Y) - (X, Y), renk
```

Örnek:

```
LINE (80, 50) - (250, 150), 2
```

Belirlenen ilk nokta (80, 50 koordinatları) çizginin başladığı noktadır. Diğer nokta ise (250,150) çizginin bittiği noktadır. Qbasic, verilen bu iki noktayı birleştirerek çizgiyi çizecektir. Örnekte geçen 2 değeri, çizginin yeşil olmasını sağlar.

LINE deyiminin hangi ekran modu ile kullanıldığı önemlidir. SCREEN 2 modunda ekranın boyutları 640x200, SCREEN 1 modunda ise 320x200 'dür. Buna göre SCREEN 1 modunda ekranın sol üst köşesi ile sağ alt köşesini birleştirecek çizgi,

```
LINE (0, 0) - (320, 200)
```

dizilimidir. Aynı işlem SCREEN 2 modunda ise,

```
LINE (0, 0) - (640, 200)
```

dizilimi ile yaptırılır.


Aşağıdaki örnek program LINE deyiminin bir uygulamasıdır. Program, ekrana büyük bir M harfi çiziyor. Programın yazımı bittikten sonra, siz de kendi isminizin baş harfini ekrana çizgilerle yazan programı geliştiriniz.

1. program

```
REM LINE deyimi uygulaması ekrana M harfi çizer
CLS
SCREEN 12
LINE (50, 50) - (50, 250): REM M nin sol bacağı
LINE (50, 50) - (250, 250): REM sola yatık çizgi
LINE (250, 250) - (450, 50): REM sağa yatık
LINE (450, 50) - (450, 250): REM M nin sağ bacağı
END
```



Programınızı **line.bas** olarak kaydediniz.



● CIRCLE Deyimi

Ekrana daire ve elips çizmek için kullanılır.

Dizilim:

CIRCLE (X,Y),yarı çap, renk, başlangıç, bitiş , elips oranı

Örnek:

```
SCREEN 12
CIRCLE (320,240),80,4
```

dairenin merkezi - x dairenin merkezi - y renk dairenin yarı çapı.

Dizilimde parantez içinde belirtilen değerler, dairenin merkez koordinatlarıdır. CIRCLE deyimi ile istenirse dairenin bir kısmı çizilebilir. Bu durumda dairenin başlangıç ve bitiş açılarının değerleri radyan cinsinden verilmelidir. Derece cinsinden verilen değerler de radyana çevrilmelidir.

CIRCLE deyimi ile elipsler de çizmek mümkündür. Bunun için çizilecek elipsin eksen oranları verilmelidir.

Örnek:

```
SCREEN 12
CIRCLE (250,250),100,2,,,1/2
```

Örnekte görüldüğü gibi, çizilecek şeklin başlangıç ve bitiş değerleri verilmemiş, ancak şeklin bir elips olması istenildiği için eksenlerin oranı verilmiştir. Bu gibi durumlarda, kullanılmayacak değerler yerine örnekte olduğu gibi virgöl


178 QBASIC

kullanılır. Eğer örnekte, virgüller kullanılmadan doğrudan $1/2$ değeri verilmiş olsaydı, Qbasic bunu eksenlerin oranı olarak değil, çizilecek dairenin başlangıç değeri olarak alacaktı ve bu da hataya sebep olacaktı.

Örnek:

```
PI=3.141592614
CIRCLE (320,240),80,,0,3*PI/2
```


Dizilimde dairenin başlangıç değeri olarak **0**, bitiş değeri olarak da **3*PI/2** belirlenmiştir. Buna göre çizdirilecek daire parçası şu şekilde olacaktır:



Ekrana elips çizdirmek için elipsin eksenlerinin oranları belirtilmelidir:

Örnek:

```
CIRCLE (320,240),80,12,,,2/7
```




$$a/b = 2/7$$

Aşağıdaki program, CIRCLE deyiminin bir uygulamasıdır. Bu deyim kullanılmasıyla ekrana çeşitli şekiller çizdirilmektedir.

2. program

```
REM CIRCLE deyimi uygulaması
REM ekrana çeşitli şekiller çizilir
CLS
SCREEN 12
CIRCLE (220,200),100,, , ,1 / 5
CIRCLE (260,190),100,,110*3.14/180,200*3.14/180, 3
CIRCLE (340,210),30,,-3.14,-3.14/2
END
```

 Programınızı **circle.bas** olarak kaydediniz.



Aşağıdaki program, iç içe daireler çizen bir programdır. Programı durdurmak için **Ctrl+Break** tuşları kullanılmalıdır.

3. program

```
REM CIRCLE deyimi uygulaması - İç içe daireler çizer
e01:
CLS
SCREEN 1


FOR i = 1 TO 150
  CIRCLE (160, 100), i, 1
NEXT i

FOR k = 150 TO 72 STEP -1
  CIRCLE (160, 100), k, 0
NEXT k
COLOR , 0

FOR t = 2 TO 150 STEP 3
  CIRCLE (160, 100), t, 2
NEXT t

FOR j = 150 TO 2 STEP -4
  CIRCLE (160, 100), j, 0
NEXT j
GOTO e01
```

 Programınızı **circle2.bas** olarak kaydediniz.



● PAINT Deyimi

Ekranında belirtilen kapalı bir alanın boyanmasında kullanılır.

Dizilim:

```
PAINT (X,Y),renk
```

Örnek:

```
SCREEN 12  
PAINT (220,200),6
```

(220, 200) ifadesiyle belirlenen koordinat, boyamanın başlayacağı, kapalı şeklin içerisinde kalan herhangi bir noktadır. Bu noktanın içerisinde bulunduğu kapalı alanın tümü boyanacaktır.

Aşağıdaki örnek, ekrana Türk bayrağı çizen bir uygulamadır.

4. program

```
CLS  
REM LINE, CIRCLE, PAINT deyimleri uygulaması  
REM Türk Bayrağı çizimi  
SCREEN 1:COLOR 4,1  
PI=3.141592  
CIRCLE (106,100), 53, 3, PI / 4.45, 7*PI / 3.85  
CIRCLE (119,100), 42, 3, PI / 4, 7 * PI / 4  
PAINT (60,100), 3
```

```
LINE (154,100) - (168,95) : LINE (154,100) - (168,105)
LINE (168,95) - (168,85) : LINE (168,105) - (168,114)
LINE (189,89) - (180,100) : LINE (189,111) - (180,100)
LINE (189,89) - (175,93) : LINE (189,111) - (175,107)
LINE (168,114) - (175,106) : LINE (168,85) - (175,93)
PAINT (160,100), 3
END
```

 Programınızı **bayrak.bas** olarak kaydediniz.



● PSET Deyimi

PSET deyimi, ekranda belirlenen koordinatlara nokta koymak için kullanılır.

Dizilim:

PSET (X,Y),renk

Örnek:

```
SCREEN 12
PSET (320,240),4
```

Dizilim ekranda (320,240) ile belirlenen yere bir nokta koyacaktır. Dizilimde geçen "4" değeri noktanın rengini belirler.

● PRESET Deyimi

PSET deyiminin tersine ekranda koordinatları verilen yerdeki noktayı silmek için kullanılır.

Dizilim:

```
SCREEN 12
PRESET (320,240),4
```

Örnek:

```
SCREEN 12
PSET (320,240),4
PRESET (320,240)
```

Dizilime göre ekranda önce (320, 240) ile belirtilen yere bir nokta konulacak, daha sonra PRESET deyimi ile bu nokta kaldırılacaktır. Dolayısıyla ekranda herhangi bir şey görüntülenmeyecektir

Kendiniz uygulayınız

Ekrana rastgele renkli noktalar çizen bir program yazınız.


Aşağıdaki program PSET deyiminin bir uygulamasıdır. Ekrana sinüs eğrisinin grafiğini çizer.

5. program

```
REM PSET deyimi uygulaması
REM ekrana sinüs eğrisi çizer
CLS
SCREEN 12
FOR I = 0 TO 360
X=45 + I / 1.5
Y=80 - SIN ( I / 180 * 3.141592) * 50
PSET (X,Y) , 2
NEXT
END
```



Programınızı **sinegri.bas** olarak kaydediniz.



● DRAW Deyimi

DRAW deyimi, ekrana standart şekillerin dışında şekil çizilmesinde kullanılır. Ne tür bir şekil çizileceği, DRAW deyimi ile birlikte kullanılan parametrelerle belirlenir.

Dizilim:

DRAW "parametreler"

DRAW deyimi ile kullanılacak parametreler ve işlevleri şunlardır:


Parametreler	İşlevi
Un	n birim yukarı hareket et.
Dn	n birim aşağı hareket et.
Ln	n birim sola hareket et.
Rn	n birim sağa hareket et.
En	n birim sağa-yukarı hareket et.
Fn	n birim sağa-aşağı hareket et.
Hn	n birim sola-yukarı hareket et.
Gn	n birim sola-aşağı hareket et.
An	90n birim açı dön.
TAn	çizim yönünü n derece döndür.
B	çizmeden hareket et.
Mx,y	ekranda x,y noktasına git.

Şimdi örneklerle bu parametreleri açıklayalım:

184 QBASIC

Paralelkenar çizdirmek için:

```
DRAW "R90 E50 L90 G50"
```



Ekrana basit bir üçgen çizdirmek için:

```
DRAW "R120 H60 G60"
```



Mx,y dizilimi ile **B** parametresi birlikte kullanılarak ekranda çizim yapılmadan hareket edilebilir.

Örnek:

```
DRAW "BM65,4"
```

Buna göre yan yana iki altıgen çizen bir program yazalım:

6. program

```
CLS
REM yanyana iki altıgen çizimi
REM TNA ve BM(x,y) parametreleri uygulaması
SCREEN 12
FOR j=1 TO 2
FOR i=0 to 360 STEP 60
DRAW "TA" + STR$(i) : REM açı belirleniyor
REM kenarları 60 birim altıgen çiziliyor
DRAW "R60"
NEXT i
REM ikinci altıgen için yer belirleniyor
DRAW "BM+65,+0"
NEXT j
END
```

Programınızı **altigen.bas** olarak kaydediniz.

Ekran çıktısı



Program
6

Grafik modu
SCREEN 12



M(x,y) parametresinin, kursörü ekranın (x,y) noktasına götürdüğünü belirttik. Bu parametre, 5. programda olduğu gibi M(+x,+y) biçiminde de kullanılabilir. Bu durumda kursör bulunduğu noktadan (+x, +y) birim ötelenir.

● POINT Deyimi

POINT deyimi, belirtilen bir noktanın renginin elde edilmesinde kullanılır.

Dizilim:

```
POINT (X,Y)
```

Örnek:

```
IF POINT(200,200)=1 THEN PRINT "mavi!!"
```

10.2 Qbasic'te Ses

Qbasic'te, çok kaliteli olmamakla birlikte çeşitli sesler üretilebilir. Bu kısımdaki uygulamalar için ses kartına ihtiyacımız olmayacaktır. Sesler PC hoparlöründen alınacaktır.

● BEEP Deyimi

İsminden de anlaşılacağı gibi, beep deyimi programda istenen yerde "bip" sesi elde etmek için kullanılır. Bu deyim, kullanıcıyı uyarmak için kullanılabilir.

Dizilim:

```
BEEP
```

Örnek:

```
DO
BEEP
LOOP UNTIL INKEY$=CHR$(27)
```

ESC tuşuna basılmadığı sürece sürekli **bip** sesi üretilecektir.

● SOUND Deyimi

Belli frekansta ve sürede ses üretmek için kullanılır.

Dizilim:

```
SOUND frekans,süre
```

Örnek:

```
SOUND 2000, 90
```

SOUND deyimi ile kullanılan değerlerden ilki, üretilecek sesin **frekansını**, diğeri ise **süresini** gösterir.

SOUND deyiminde, süre için kullanılan birim, saniyenin yaklaşık olarak 18' de biridir. Diğeri bir deyişle, yaklaşık 1 saniyelik ses üretmek için süre değeri 18 olarak verilmelidir. Kullanılacak değer 0 (sıfır) ile 65.535 arasında olabilir.

Frekans değeri olarak belirlenecek değer 37 ile 32.767 arasında olmalıdır. Frekans değeri arttıkça ses tizleşir. Belli bir değerden sonrasını ise insan kulağı algılayamaz.

Aşağıdaki program kuş sesi üretir:

7. program

```
CLS
REM SOUND deyimi uygulaması
REM kuş sesi üretir
PRINT "çıkış için ESC.."
DO
  FOR i = 3000 TO 4500 STEP 175
    SOUND i, 0.3
  NEXT
LOOP UNTIL INKEY$=CHR$(27)
END
```



Programınızı **kussesı.bas** olarak kaydediniz.

Rastgele frekans değerleri üretilerek de ilginç sesler elde edilebilir:

Örnek:

```
DO
SOUND RND*1000+37.2, 2
LOOP
```

Aşağıdaki program, klavyenizi bir müzik klavyesi yapacaktır.

8. program

```
CLS
REM SOUND deyimi uygulaması
REM müzik klavyesi
PRINT "tuşlara basın!!"
```

```

PRINT "yıldız (*) işareti son"
LOCATE 8,1
e01:
K$=INKEY$
IF K$="" THEN GOTO e01:
IF K$="*" THEN END
IF ASC(K$) < 33 THEN GOTO e01

PRINT CHR$ (14) ; CHR$ (32) ;
SOUND ASC (K$)* 10, 3
GOTO e01
END

```

 Programınızı **muzik.bas** olarak kaydediniz.

SOUND deyimi ile ilgili bir başka örnek de aşağıdaki programdır. Programda İstiklâl Marşı çalınmaktadır. Bu programı yazarken, özellikle data satırındaki değerlere çok dikkat ediniz. Virgül karakterlerinin yerini şaşırmayınız ve ondalık sayılardaki nokta karakteri ile karıştırmayınız.

9. program

```

REM **** TÜRK MİLLİ MARŞI ****
SAYI = 0
DIM TS(70, 2)
ETIKET1:
L = 1:NT = 65
IF SAYI = 2 GOTO ETIKET2
FOR J = 1 TO NT
READ TS(J, 1), TS(J, 2)
NEXT J
FOR J = 1 TO NJ
READ JS(J, 1), JS(J, 2)
NEXT J
FOR J = 1 TO NT
IF TS(J, 1) <> 32767 THEN S = 2 * TS(J, 1) ELSE S = TS(J, 1)
SOUND S, TS(J, 2) * L
NEXT
SAYI = SAYI + 1
RESTORE
GOTO ETIKET1
ETIKET2:
END
DATA 246.94,12,329.63,12,371.23,12,392.0,12,311.66,9
DATA 371.23,3,329.63,32,32767,4,329.63,12

```

```

DATA 440.0,12,493.88,12,523.25,9,493.88,3,422.0,9,493.88,3,440.0
DATA 32,32767,4,493.88,4,467.0,4,493.88,4
DATA 371.23,10,32767,2,371.23,20,440.0,3,392.0,9,311.99,3,329.63
DATA 9,371.23,3,392.0,9,440.0,3,493.88,9,523.25,3,587.33,9,659.26,3
DATA 587.33,12,293.66,4,277.63,4,293.66,4,493.88,12,440.0,12,392.0
DATA 32,32767,4,246.94,4,234.0,4,246.94,4
DATA 371.23,12,246.94,12,493.88,9,440.0,3,392.0,4,371.23,4
DATA 392.0,4,329.63,12,659.26,21,587.33,3,523.25,9,493.88,3
DATA 440.0,9,392.0,3,371.23,9,329.63,3,493.88,12,246.94,12
DATA 329.63,20,32767,4

```



Programınızı **istmars.bas** olarak kaydediniz.

● PLAY Deyimi

String dizi içerisindeki notalara ait sesleri üretmek için kullanılır.

Dizilimi:

PLAY "String dizi"

Örnek:

PLAY "EABBE#O3G#"

Verilen diziye göre, istenilen şartlarda notaları çalar. String dizi içerisinde birçok özel karakter bulunur. Bu karakterlerden, notaları temsil edenler, oktav seçen, hızlı ve yavaş giriş yapanlar mevcuttur. Bu karakterler kısaca şöyledir:

A (la), **B**(si), **C**(do), **D**(re), **E**(mi), **F**(fa), **G**(sol) karakterleri yanlarında belirtilen notaları temsil eder. Bulunan oktavdaki gamın notalarını çalar. Harflerden sonra gelen "**#**" ve "**+**" işareti hızlı girişi, "**-**" ise yavaş girişi belirler.

On : Oktav seçmek için kullanılır. 0-6 arasında 7 değişik oktav vardır.

Nn : İstenilen notayı çalmak için kullanılır. 7 oktav içerisinde 84 nota vardır. N değeri 0 ile 84 arasında değişir. <A-G> karakterleri yerine kullanılabilir.

Ln: Kendisinden sonra gelecek olan notanın uzunluğunu belirler. Tam nota uzunluğunun 1/n kadarını gösterir. Uzunluk, 1 ile 64 birim arasında değişir.

Tn: Tempo ayarlaması yapar. 32-255 arası bir değer alabilir.

MF: Kendisinden sonra gelen notaların ön plânda çalınacağını gösterir.

MB: Fon müziği çalmak için kullanılır. Bilgisayar çalışmasına devam ederken müzik ara belleğindeki notaları çalar. Her defasında 32 nota çaldırabilir.

X değişkeni: Bir değişken ismi ile bir dizi notanın çalınması işlemini yapar.

P (Pause): Bu karakterin geçtiği yerde duraklanır. 1-64 arası bir değer alabilir. P1, P2 gibi.

Örnek olarak "Bir ilkbahar sabahı..." isimli şarkıyı seslendirecek bir program yazalım. Programı yazarken "0" (sıfır) ile "O" (oktav-O harfi) karakterlerini birbirine karıştırmayınız ve karakterler arasında boşluk bırakmayınız.

10. program

```
REM PLAY deyimi uygulaması
REM Bir ilkbahar sabahı...

PLAY "L4O4GG#GFA#G#L2G"
PLAY "L4GGGFGA#GG#"
PLAY "L4G#G#G#G#A#O5CL2D"
PLAY "L4O5D#DCO4A#A#O5CO4G#GP8"
PLAY "O5CCCCO4A#O5DL2C"
PLAY "L4O5DD#FD#CDO4A#A#"
PLAY "O5D#DCO4A#G#GL2G#"
PLAY "L4A#O5CDCO4A#O5CO4G#L2G"
PLAY "L4O5CCCCO4A#O5DCDD#FD#DCO4A#"
PLAY "O5D#DCO4A#G#GL2G#"
PLAY "L4GG#GFD#DL1C"
END
```



Programınızı **ilkbahar.bas** olarak kaydediniz.